Создает внутренний документ из внутреннего, создает связь между документами, копирует файлы и значения доп. реквизитов
//Создание Обращение в техподдержку на основании выхода сотрудника
Предмет = Процесс.Предметы[0].Предмет;
ОбращениеВТехподдержку = Справочники.ВнутренниеДокументы.СоздатьЭлемент();
ОбращениеВТехПоддержку.ВидДокумента = lvv_ВызовСервера.lvv_ПолучитьЗначениеДопПараметра("ВидДокументаОбращениеВТехПоддержку");
ОбращениеВТехПоддержку.Заполнить(Предмет);
ОбращениеВТехПоддержку.Заголовок = "Выход сотрудника: "+Предмет.Заголовок;
ОбращениеВТехПоддержку.Создал = Предмет.Подготовил;
ОбращениеВТехПоддержку.Подразделение = Предмет.Подразделение;
ОбращениеВТехПоддержку.Подготовил = Предмет.Подготовил;
ОбращениеВТехПоддержку.Ответственный = Предмет.Подготовил;
ОбращениеВТехПоддержку.Папка = lvv_ВызовСервера.lvv_ПолучитьЗначениеДопПараметра("ПапкаОбращенияВТехподдержку");
ОбращениеВТехПоддержку.Записать();
//РаботаСРабочимиГруппами.ПерезаполнитьРабочуюГруппуДокумента(ОбращениеВТехПоддержку.Ссылка, Истина)


//Установка связи между документами 
СвязиДокументов.УстановитьСвязь(
					Предмет.Ссылка,
					Неопределено,
					ОбращениеВТехПоддержку.Ссылка,
					lvv_ВызовСервера.lvv_ПолучитьЗначениеДопПараметра("СвязьОснованиеОбращенияВТехподдержку"));

//Копирование файлов из документа основания с проверкой на расширения файла					
Запрос = Новый Запрос;
Запрос.Текст =
"ВЫБРАТЬ
|Файлы.Ссылка
| ИЗ
|  Справочник.Файлы КАК Файлы
| ГДЕ
|  Файлы.ВладелецФайла = &ВладелецФайла
|И Файлы.ПометкаУдаления = ЛОЖЬ";

Запрос.УстановитьПараметр("ВладелецФайла", Предмет.Ссылка);

РезультатЗапроса = Запрос.Выполнить();

ВыборкаДетальныеЗаписи = РезультатЗапроса.Выбрать();

Пока ВыборкаДетальныеЗаписи.Следующий() Цикл
    ФайлОснование = ВыборкаДетальныеЗаписи.Ссылка;
    Если СтрНайти("BMP GIF JPG JPEG PNG",ВРЕГ(ФайлОснование.ТекущаяВерсияРасширение))>0 Тогда
   	    ФайлСсылка = РаботаСФайламиВызовСервера.СкопироватьФайл(ФайлОснование, ОбращениеВТехПоддержку.Ссылка);
   	КонецЕсли;   	    
КонецЦикла;

// Перененос свойств из документа основания в обращение в техподдержку
Запрос = Новый Запрос;
Запрос.Текст = 
	"ВЫБРАТЬ
	|	ВнутренниеДокументыДополнительныеРеквизиты.Свойство КАК Свойство,
	|	ВнутренниеДокументыДополнительныеРеквизиты.Значение КАК Значение,
	|	ВнутренниеДокументыДополнительныеРеквизиты.ТекстоваяСтрока КАК ТекстоваяСтрока,
	|	ВнутренниеДокументыДополнительныеРеквизиты.Ссылка КАК Ссылка
	|ИЗ
	|	Справочник.ВнутренниеДокументы.ДополнительныеРеквизиты КАК ВнутренниеДокументыДополнительныеРеквизиты
	|ГДЕ
	|	ВнутренниеДокументыДополнительныеРеквизиты.Ссылка = &Ссылка"
;
Запрос.УстановитьПараметр("Ссылка",Предмет.Ссылка);

Выборка = Запрос.Выполнить().Выбрать();
МассивСтруктур = Новый Массив;
Пока Выборка.Следующий() Цикл
	МассивСтруктур.Добавить(Новый Структура("Свойство, Значение",Выборка.Свойство, Выборка.Значение));	
КонецЦикла;
УправлениеСвойствами.ЗаписатьСвойстваУОбъекта(ОбращениеВТехПоддержку.Ссылка, МассивСтруктур);
